ABSTRACT

There is an ever-increasing number of study designs and analyses of clinical trials using Bayesian frameworks to interpret the treatment effects. Many research scientists prefer to understand the power and probability of taking a new drug forward across the whole range of possible true treatment effects, rather than focusing on one particular value to power the study. Examples will be used in this paper to show how to compute Bayesian probabilities using the SAS/STAT procedures PROC MIXED and PROC UNIVARIATE. Particular emphasis is given to the application on efficacy analysis, including the comparison of new drug to placebo and to a standard drug on the market.

INTRODUCTION

A Bayesian framework in early phase drug development is based on the idea that the new drugs not only might do something but also how they are compared to the standard drug on the market. It will help us to make a correct decision before we start a study rather than making up the decision at the end of the study. The example in this paper will explain how to set up a Bayesian framework on an efficacy endpoint. The first critical success factor is to show separation from placebo, as in a standard power calculation. The second critical success factor is to give confidence that the new drug is not inferior to the standard drug.

SETTING UP A BAYESIAN FRAMEWORK ON AN EFFICACY ENDPOINT

The choice to set up Bayesian criteria usually depends on your knowledge about the compound, your clinical plan and the competition. In some therapeutic areas, the efficacy should be better than standard of care. In other therapeutic areas, the efficacy is equivalent with the standard of care, but the new drug has a better safety profile and less side effects.

In this example, two critical success factors will be used for efficacy at the end of the study. The criteria are based on a Bayesian interpretation of the results using non-informative priors. They are:

Criterion 1: At least 90% probability that new drug has an improvement compared to placebo for the efficacy endpoint.

Criterion 2: At least 80% probability that new drug is not inferior to the standard drug (or drug on the market) by 5 units or more for the efficacy endpoint.

Criterion 1 is used to represent evidence of efficacy in the study. Criterion 2 will aid internal decision making on future development of the compound in the indication of this therapeutic area.

DATA SIMULATION

Suppose we have simple clinical trial data with a sample size of 90 subjects randomly assigned to placebo, a new drug and a standard drug. The outcome is efficacy measured at pre and post dose. The first measurement is a pre-dose baseline assessment which is followed by a post-dose measurement. The following code produces this simulated efficacy data:

```/* introduce variability */

data sample;
  do subjid=1 to 90;
    if uniform(12345)<=0.35 then treatment=1;
    else if 0.35<uniform(12345)<0.8 then treatment=2;
    else treatment=3;
    baseline= 100+12.5*(normal(2345));
    result=90+12.5*(normal(6789));
    if treatment>1 then result=result-5;
    change=result-baseline;
    output;
  end;
run;```
STEP1. FIT ANCOVA MODEL USING PROC MIXED WITH PRIOR STATEMENT OUTPUT (POST) DATA SET

The primary comparison of interest is the difference between the new drug and placebo, followed by the difference between the new drug and standard drug. The comparison of standard drug to placebo will also be obtained. When we apply a simple ANCOVA MODEL on the simulated efficacy data, the SAS/STAT procedure PROC MIXED provides options to output a dataset containing the posterior sample. This Bayesian inference is justified on the assumption of non-informative priors for all parameters in the model.

Example SAS code for step 1 is given below:

```sas
proc mixed data=sample method=ml;
   class treatment;
   prior / out=post nsample=10000 seed=5969953;
   model change=baseline treatment / ddfm=kr;
   lsmeans treatment / cl pdiff;
run;
quit;
```

STEP2. PREPARE DATA FOR PROC UNIVARIATE

With the output dataset (post) from PROC MIXED above, we need to prepare the data before we can use PROC UNIVARIATE to analyse the posterior sample. Suppose we have treatment 1 as placebo, treatment 2 as the new drug and treatment 3 as the standard drug, dif1 in the post dataset as the difference between the new drug and placebo, dif2 as the difference between the standard drug and placebo, and dif3 as the difference between the new drug and standard drug. It is necessary that we reverse the sign of the difference for dif1 and dif2 because SAS reference higher order. Next, we need to create 2 sets of probability variables according the criteria setup in the Bayesian framework.

Example SAS code for step 2 is given below:

```sas
data post2(drop=i);
   set post(keep=dif1-dif3 lsm1-lsm3);
   dif1=-dif1;
   dif2=-dif2;
   array dif{3} dif1-dif3;
   array p0{3} p01-p03;
   array p5{3} p51-p53;
   do i=1 to 3;
      if not missing(dif{i}) then do;
         if dif{i}<0 then p0{i}=1; else p0{i}=0;
         if dif{i}<5 then p5{i}=1; else p5{i}=0;
      end;
   end;
run;
```

STEP3. GET BAYESIAN PROBABILITIES USING PROC UNIVARIATE

With the new data (post2), we can use PROC UNIVARIATE to get descriptive statistics and Bayesian probabilities including the posterior probability that the difference of new drug and standard drug to placebo, and the posterior probability that the difference of new drug to standard drug is <= 5 units.

Example SAS code for step 3 is given below:

```sas
*-----create param data set to loop thru indv macro -----*;
proc transpose data=post2(obs=1) out=param(keep=_name_ rename=(_name_=param));
run;

%macro sum(param);
   proc univariate data=post2 noprint;
      var &param.;
      output out=&param. mean=m1 median= me1 std=s1 pctlpts = 2.5 97.5 pctlpre=P;
   run;
```
Bayesian Framework in Early Phase Drug Development with SAS® Example, continued

data &param.;
  set &param.;
  var="&param.";
run;
%mend sum;

data _null_;  
  set param;
  call execute('%nrstr(%sum(%str('||param||'))'));
run;

STEP4. REARRANGE DATA SET TO CREATE FINAL REPORT

This step is simply to rearrange the data sets for PROC REPORT.

Example SAS code for final step 4 is given below:

data panel_l(drop=var);
  length param $ 60;
  retain param;
  set lsm1-lsm3 dif1-dif3 ;
  if var='lsm1' then do; order=1; param='Placebo';end;
  if var='lsm2' then do; order=2; param='New Drug';end;
  if var='lsm3' then do; order=3; param='Standard Drug';end;
  if var='dif1' then do; order=4; param='New Drug - Placebo';end;
  if var='dif2' then do; order=5; param='Standard Drug - Placebo';end;
  if var='dif3' then do; order=6; param='New Drug - Standard Drug';end;
run;

data panel_r(keep=order m1 x);
  set p01-p03 p51-p53;
  if var in ('p01','p51') then order=4;
  else if var in ('p02','p52') then order=5;
  else if var in ('p03','p53') then order=6;
  if index(var,'p0')>0 then x=0;
  else if index(var,'p5')>0 then x=5;
run;

proc sort data=panel_r;
  by order;
run;

proc transpose data=panel_r out=panel_tr(drop=_name_ _label_);
  by order;
  var m1;
  id x;
run;

data final;
  merge panel_l panel_tr;
  by order;
run;
The data set FINAL above would have the following report by PROC REPORT:

<table>
<thead>
<tr>
<th>Treatment effect/contrast</th>
<th>Mean</th>
<th>SD</th>
<th>Median</th>
<th>2.5%</th>
<th>97.5%</th>
<th>0</th>
<th>5</th>
</tr>
</thead>
<tbody>
<tr>
<td>Placebo</td>
<td>-12.727</td>
<td>2.351</td>
<td>-12.752</td>
<td>-17.261</td>
<td>-8.117</td>
<td>0.984</td>
<td>0.999</td>
</tr>
<tr>
<td>New Drug</td>
<td>-19.291</td>
<td>2.051</td>
<td>-19.299</td>
<td>-23.328</td>
<td>-15.229</td>
<td>0.765</td>
<td>0.990</td>
</tr>
<tr>
<td>Standard Drug</td>
<td>-15.067</td>
<td>2.146</td>
<td>-15.051</td>
<td>-19.302</td>
<td>-10.893</td>
<td>0.922</td>
<td>0.998</td>
</tr>
<tr>
<td>New Drug - Placebo</td>
<td>-6.564</td>
<td>3.113</td>
<td>-6.525</td>
<td>-12.744</td>
<td>-0.491</td>
<td>0.984</td>
<td>0.999</td>
</tr>
<tr>
<td>Standard Drug - Placebo</td>
<td>-2.341</td>
<td>3.198</td>
<td>-2.344</td>
<td>-8.684</td>
<td>3.966</td>
<td>0.765</td>
<td>0.990</td>
</tr>
<tr>
<td>New Drug – Standard Drug</td>
<td>-4.223</td>
<td>2.981</td>
<td>-4.232</td>
<td>-10.062</td>
<td>1.793</td>
<td>0.922</td>
<td>0.998</td>
</tr>
</tbody>
</table>

Table1. Report Including the Bayesian Probability on Efficacy

CONCLUSION

Bayesian probability has become more useful in the study design for research scientists in early phase drug development, because it provides a straightforward statistical framework that not only helps the study team to communicate and understand the new drug, but also enables the management team to make the correct decisions towards a viable pipeline and optimal marketing candidates. It was the intention in this paper and the given example to present Bayesian probabilities to a wide audience in a comprehensive manner and has it be a useful reference for statistical analysts and others who perform statistical analysis duties in clinical trials or clinical research.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Howard Liang
Address: 12468 Independence Ave
City, State ZIP: Savage, MN 55344
Work Phone: (952) 496-1577
E-mail: hliang5995@yahoo.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.